Blending Problem

Decision Variables:

M_j= quantity of each metal to be produced (j=1,2,3)
R_r= quantity of each type raw material (r=1,2)
S_s= quantity of each type of scrap (s=1,2)
I_i = quantity of ingot (i=1)
P_j = quantity of each of the pure metal (j=1,2,3)

Therefore,there are 11 decision variables from which all of them take float data except I_{ingos} integer data,thus, this problem is a mixed-integer linear problem.

Other Variables:

Cost variables:
CostMetal[j]= cost of j^{th} pure metal (j=1,2,3)
CostRaw[r]=cost of each r^{th} raw materials (r=1,2)
CostScrap[s]= cost of s^{th} scrap (s=1,2)
CostIngo[i]= cost of i^{th} ingots (i=1)

variables for production constraints:
Low[j] = gives lower bounds on the quantity of j^{th} metal in the alloy
Upper[j] = gives upper bounds on the quantity of j^{th} metal in the alloy
PR[j,r] = proportion of j^{th} metal in r^{th} raw material
PS[j,s] = proportion of j^{th} metal in s^{th} scrap
PI[j,I] = proportion of j^{th} metal in i^{th} ingots
Alloy=71

Objective Function:

The goal is to minimize the cost while blending the sources into the alloy satisfying the production constraints

Minimize

$$\text{Cost} = \sum_{j=1}^{3} CostMetal(j) * P(j) + \sum_{r=1}^{2} CostRaw(r) * R(r) + \sum_{s=1}^{2} CostScrap(s) * S(s) + \sum_{i=1}^{1} CostIgnos(i) * I(i)$$

= 22*P1+10*P2+13*P3 + 6*R1 + 5*R2 + 7*S1+ 8*S2 + 9*I1

Constraints:

total production constraint:
M1+M2+M3=71

source constraints:

$$M(j) = P(j) + \sum_{k=1}^{2} PR[j,k] * R(k) + \sum_{k=1}^{2} PS[j,k] * S(k) + \sum_{k=1}^{1} PI[j,k] * I(k)$$

Where,j=1,2,3

That is:
P1+(.2*R1+.01*R2)+(0*S1+.01*S2)+.1*I = M1
P2+(.05*R1+0*R2)+(.6*S1+0*S2)+.45*I = M2
P3+(.05*R1+.3*R2)+(.4*S1+.7*S2)+.45*I = M3

metal production constraints:
$Low(j) * Alloy \leq M(j) \leq Upper(j) * Alloy$
Where,j =1,2,3

That is:
$.05*71 \leq M1 \leq .1*71$
$.3*71 \leq M2 \leq .4*71$
$.6*71 \leq M3 \leq .8*71$

$P_j \geq 0$, $R_{raws} \geq 0$, $S_{scraps} \geq 0$, $I_{ingos} \geq 0$

OPL code:

We write the OPL code from the above model to solve this Linear Programming problem :

OPL Model:

```
//constant variables
int   NbMetals = ...;
int   NbRaw = ...;
int   NbScrap = ...;
int   NbIngo = ...;


//data structures to define data storage for variables
range Metals = 1..NbMetals; //range of integers from 1 to number of metals
range Raws = 1..NbRaw;
range Scraps = 1..NbScrap;
range Ingos = 1..NbIngo;

//cost variables
float CostMetal[Metals] = ...;//array to store metal´s cost indexed over
all the metal types
float CostRaw[Raws] = ...;
float CostScrap[Scraps] = ...;
float CostIngo[Ingos] = ...;

//array for production constraints
float Low[Metals] = ...;
float Up[Metals] = ...;

//array for source constraints
float PercRaw[Metals][Raws] = ...;
float PercScrap[Metals][Scraps] = ...;
float PercIngo[Metals][Ingos] = ...;

//total  production constraint
int Alloy  = ...;

//define dvar='decision variable'
dvar float+    p[Metals]; //p takes positive float values
```

```
dvar float+    r[Raws];
dvar float+    s[Scraps];
dvar int+      i[Ingos]; //integer quantity of ignot
dvar float    m[j in Metals] in Low[j] * Alloy .. Up[j] * Alloy;//specifing
the range for m

//model the problem
minimize
  sum(j in Metals) CostMetal[j] * p[j] +
  sum(j in Raws)   CostRaw[j]   * r[j] +
  sum(j in Scraps) CostScrap[j] * s[j] +
  sum(j in Ingos)  CostIngo[j]  * i[j];
subject to {
  forall( j in Metals )
    ct1: //source constraint
      m[j] ==
      p[j] +
      sum( k in Raws )   PercRaw[j][k] * r[k] +
      sum( k in Scraps ) PercScrap[j][k] * s[k] +
      sum( k in Ingos )  PercIngo[j][k] * i[k];
    ct2:  //total production constraint
      sum( j in Metals ) m[j] == Alloy;
}
```

OPL Data:

```
//data for index
NbMetals = 3; //three metals
NbRaw = 2; //two raw materials
NbScrap = 2; //two kinds of scrap
NbIngo = 1;  //one kind of ingot


//cost data
CostMetal = [22, 10, 13];//cost of three types of pure metals
CostRaw = [6, 5]; //per kilo cost of two types of raw materials
CostScrap = [ 7, 8];//per kilo cost of two types of Scrap
CostIngo = [ 9 ]; //per kilo cost of Ingot

//production constraints on the quantity of each type of metal in the alloy
Low = [0.05, 0.30, 0.60];
Up = [0.10, 0.40, 0.80];

//source constraints on the proportions of each of the metals
PercRaw = [ [ 0.20, 0.01 ], [ 0.05, 0 ], [ 0.05, 0.30 ] ];
PercScrap = [ [ 0 , 0.01 ], [ 0.60, 0 ], [ 0.40, 0.70 ] ];
PercIngo = [ [ 0.10 ], [ 0.45 ], [ 0.45 ] ];


//total  production constraint
Alloy  = 71;
```

Solution:

Result of the objective function:Minimum cost is **653.61**

Result of the decision variables:
Amount of pure metals: **P = [0.046667 ,0 ,0]**
Amount of Raw Materials: **R = [0 ,0]**
Amount of Scraps: **S = [17.417 ,30.333]**
Amount of Ingots: **I = [32]**
Amount of Metals to be Produced : **M = [3.55, 24.85 ,42.6]**